Computing the Coiflet C6-filter

Ansatz

In[1]:=

```
ansatz = (1/2 + 1/4 * (z + 1/z)) * (1 + (1/2 - 1/4 * (z + 1/z)) * (a[0] + a[1] * z))
```

Out[1]=

$$\left(\frac{1}{2}+\frac{1}{4}\left(\frac{1}{z}+z\right)\right)\left(1+\left(\frac{1}{2}+\frac{1}{4}\left(-\frac{1}{z}-z\right)\right)(a[0]+z\,a[1])\right)$$

In[2]:=

```
Expand[ansatz]
```

Out[2]=

$$\frac{1}{2}+\frac{1}{4z}+\frac{z}{4}+\frac{a[0]}{8}-\frac{a[0]}{16z^2}-\frac{1}{16}z^2 a[0]-\frac{a[1]}{16z}+\frac{1}{8}z\,a[1]-\frac{1}{16}z^3 a[1]$$

In[3]:=

```
h = CoefficientList[% z^2, z] Sqrt[2]
```

Out[3]=

$$\left\{-\frac{a[0]}{8\sqrt{2}},\ \sqrt{2}\left(\frac{1}{4}-\frac{a[1]}{16}\right),\ \sqrt{2}\left(\frac{1}{2}+\frac{a[0]}{8}\right),\ \sqrt{2}\left(\frac{1}{4}+\frac{a[1]}{8}\right),\ -\frac{a[0]}{8\sqrt{2}},\ -\frac{a[1]}{8\sqrt{2}}\right\}$$

Orthogonality conditions

In[4]:=

```
eq1 = Sum[h[[k]]^2, {k, 1, 6}] == 1
```

Out[4]=

$$2\left(\frac{1}{2}+\frac{a[0]}{8}\right)^2+\frac{a[0]^2}{64}+2\left(\frac{1}{4}-\frac{a[1]}{16}\right)^2+2\left(\frac{1}{4}+\frac{a[1]}{8}\right)^2+\frac{a[1]^2}{128}==1$$

In[5]:=

```
eq1 = Map[Expand[64 #] &, %]
```

Out[5]=

$$48+16\,a[0]+3\,a[0]^2+4\,a[1]+3\,a[1]^2==64$$

In[6]:=

```
eq2 = Sum[h[[k]] * h[[k + 2]], {k, 1, 4}] == 0
```

Out[6]=

$$-\frac{1}{4}\left(\frac{1}{2}+\frac{a[0]}{8}\right)a[0]+2\left(\frac{1}{4}-\frac{a[1]}{16}\right)\left(\frac{1}{4}+\frac{a[1]}{8}\right)-\frac{1}{8}\left(\frac{1}{4}+\frac{a[1]}{8}\right)a[1]==0$$

In[7]:=

```
eq2 = Map[Expand[32 #] &, %]
```

Out[7]=

$$4-4\,a[0]-a[0]^2-a[1]^2==0$$

In[8]:=

```
eq3 = Sum[h[[k]] * h[[k + 4]], {k, 1, 2}] == 0
```

Out[8]=

$$\frac{a[0]^2}{128}-\frac{1}{8}\left(\frac{1}{4}-\frac{a[1]}{16}\right)a[1]==0$$

In[9]:=
```
eq3 = Map[Expand[128 #] &, %]
```
Out[9]=
$$a[0]^2 - 4\,a[1] + a[1]^2 == 0$$

Obtaining the solution

In[10]:=
```
sol = Solve[{eq1, eq2, eq3}, {a[0], a[1]}]
```
Out[10]=
$$\left\{\left\{a[0] \to \frac{1}{2}\left(-1-\sqrt{7}\right),\ a[1] \to \frac{1}{2}\left(3+\sqrt{7}\right)\right\},\ \left\{a[0] \to \frac{1}{2}\left(-1+\sqrt{7}\right),\ a[1] \to \frac{1}{2}\left(3-\sqrt{7}\right)\right\}\right\}$$

In[11]:=
```
c6coeffs = h /. sol[[2]]
```
Out[11]=
$$\left\{-\frac{-1+\sqrt{7}}{16\sqrt{2}},\ \sqrt{2}\left(\frac{1}{4}+\frac{1}{32}\left(-3+\sqrt{7}\right)\right),\ \sqrt{2}\left(\frac{1}{2}+\frac{1}{16}\left(-1+\sqrt{7}\right)\right),\ \sqrt{2}\left(\frac{1}{4}+\frac{1}{16}\left(3-\sqrt{7}\right)\right),\ -\frac{-1+\sqrt{7}}{16\sqrt{2}},\ -\frac{3-\sqrt{7}}{16\sqrt{2}}\right\}$$

In[12]:=
```
N[c6coeffs]
```
Out[12]=
```
{-0.0727326, 0.337898, 0.852572, 0.384865, -0.0727326, -0.0156557}
```

Frequency representation

In[13]:=
```
C6[ω_] := Sum[c6coeffs[[k]] Exp[I ω (k - 3)], {k, 1, 6}]
```
In[14]:=
```
Plot[Abs[C6[ω]], {ω, 0, Pi}, Filling → Axis]
```
Out[14]=


Direct computation of C6

In[15]:=
```
Clear[h]
```

In[16]:=
```
hpol[z_] = Sum[h[k] z^k, {k, -2, 3}]
```
Out[16]=

$$\frac{h[-2]}{z^2} + \frac{h[-1]}{z} + h[0] + z\,h[1] + z^2\,h[2] + z^3\,h[3]$$

Orthogonality conditions

In[17]:=
```
o1 = Sum[h[k]^2, {k, -2, 3}] == 1
```
Out[17]=

$$h[-2]^2 + h[-1]^2 + h[0]^2 + h[1]^2 + h[2]^2 + h[3]^2 == 1$$

In[18]:=
```
o2 = Sum[h[k] * h[k + 2], {k, -2, 1}] == 0
```
Out[18]=

$$h[-2]\,h[0] + h[-1]\,h[1] + h[0]\,h[2] + h[1]\,h[3] == 0$$

In[19]:=
```
o3 = Sum[h[k] * h[k + 4], {k, -2, -1}] == 0
```
Out[19]=

$$h[-2]\,h[2] + h[-1]\,h[3] == 0$$

Low-pass conditions

In[20]:=
```
t1 = hpol[1] == Sqrt[2]
```
Out[20]=

$$h[-2] + h[-1] + h[0] + h[1] + h[2] + h[3] == \sqrt{2}$$

In[21]:=
```
t2 = hpol[-1] == 0
```
Out[21]=

$$h[-2] - h[-1] + h[0] - h[1] + h[2] - h[3] == 0$$

In[22]:=
```
t3 = hpol'[1] == 0
```
Out[22]=

$$-2\,h[-2] - h[-1] + h[1] + 2\,h[2] + 3\,h[3] == 0$$

In[23]:=
```
t4 = hpol'[-1] == 0
```
Out[23]=

$$2\,h[-2] - h[-1] + h[1] - 2\,h[2] + 3\,h[3] == 0$$

Solution

In[24]:=
```
Solve[{o1, o2, o3, t1, t2, t3, t4}, Table[h[k], {k, -2, 3}]]
```

Out[24]=

$$\left\{\left\{h[-2]\to\frac{1}{32}\left(\sqrt{2}-\sqrt{14}\right),\ h[-1]\to\frac{1}{32}\left(5\sqrt{2}+\sqrt{14}\right),\ h[0]\to\frac{1}{16}\left(7\sqrt{2}+\sqrt{14}\right),\ h[1]\to\frac{1}{16}\left(7\sqrt{2}-\sqrt{14}\right),\ h[2]\to\frac{1}{32}\left(\sqrt{2}-\sqrt{14}\right),\ h[3]\to\frac{1}{32}\left(-3\sqrt{2}+\sqrt{14}\right)\right\},\right.$$
$$\left.\left\{h[-2]\to\frac{1}{32}\left(\sqrt{2}+\sqrt{14}\right),\ h[-1]\to\frac{1}{32}\left(5\sqrt{2}-\sqrt{14}\right),\ h[0]\to\frac{1}{16}\left(7\sqrt{2}-\sqrt{14}\right),\ h[1]\to\frac{1}{16}\left(7\sqrt{2}+\sqrt{14}\right),\ h[2]\to\frac{1}{32}\left(\sqrt{2}+\sqrt{14}\right),\ h[3]\to\frac{1}{32}\left(-3\sqrt{2}-\sqrt{14}\right)\right\}\right\}$$

Direct computation of the Coiflet C12 filter

In[25]:=
```
Clear[h, hpol]
```

In[26]:=
```
hpol[z_] := Sum[h[k] * z^k, {k, -4, 7} ]
```

Orthogonality conditions

In[27]:=
```
o[1] = Sum[h[k]^2, {k, -4, 7}] == 1
```

Out[27]=

$$h[-4]^2+h[-3]^2+h[-2]^2+h[-1]^2+h[0]^2+h[1]^2+h[2]^2+h[3]^2+h[4]^2+h[5]^2+h[6]^2+h[7]^2 == 1$$

In[28]:=
```
o[2] = Sum[h[k] * h[k + 2], {k, -4, 5}] == 0
```

Out[28]=

$$h[-4]\,h[-2]+h[-3]\,h[-1]+h[-2]\,h[0]+h[-1]\,h[1]+h[0]\,h[2]+h[1]\,h[3]+h[2]\,h[4]+h[3]\,h[5]+h[4]\,h[6]+h[5]\,h[7] == 0$$

In[29]:=
```
o[3] = Sum[h[k] * h[k + 4], {k, -4, 3}] == 0
```

Out[29]=

$$h[-4]\,h[0]+h[-3]\,h[1]+h[-2]\,h[2]+h[-1]\,h[3]+h[0]\,h[4]+h[1]\,h[5]+h[2]\,h[6]+h[3]\,h[7] == 0$$

In[30]:=
```
o[4] = Sum[h[k] * h[k + 6], {k, -4, 1}] == 0
```

Out[30]=

$$h[-4]\,h[2]+h[-3]\,h[3]+h[-2]\,h[4]+h[-1]\,h[5]+h[0]\,h[6]+h[1]\,h[7] == 0$$

In[31]:=
```
o[5] = Sum[h[k] * h[k + 8], {k, -4, -1}] == 0
```

Out[31]=

$$h[-4]\,h[4]+h[-3]\,h[5]+h[-2]\,h[6]+h[-1]\,h[7] == 0$$

In[32]:=
```
o[6] = Sum[h[k] * h[k + 10], {k, -4, -3}] == 0
```

Out[32]=

$$h[-4]\,h[6]+h[-3]\,h[7] == 0$$

Low-pass conditions

In[33]:=
```
t[1] = hpol[1] == Sqrt[2]
```
Out[33]=
$$h[-4]+h[-3]+h[-2]+h[-1]+h[0]+h[1]+h[2]+h[3]+h[4]+h[5]+h[6]+h[7]==\sqrt{2}$$

In[34]:=
```
t[2] = hpol[-1] == 0
```
Out[34]=
$$h[-4]-h[-3]+h[-2]-h[-1]+h[0]-h[1]+h[2]-h[3]+h[4]-h[5]+h[6]-h[7]==0$$

In[35]:=
```
H[ω_] := hpol[Exp[I ω]]
```

In[36]:=
```
t[3] = H'[0] == 0
```
Out[36]=
$$-4\,i\,h[-4]-3\,i\,h[-3]-2\,i\,h[-2]-i\,h[-1]+i\,h[1]+2\,i\,h[2]+3\,i\,h[3]+4\,i\,h[4]+5\,i\,h[5]+6\,i\,h[6]+7\,i\,h[7]==0$$

In[37]:=
```
t[3] = Map[Expand[I #] &, %]
```
Out[37]=
$$4\,h[-4]+3\,h[-3]+2\,h[-2]+h[-1]-h[1]-2\,h[2]-3\,h[3]-4\,h[4]-5\,h[5]-6\,h[6]-7\,h[7]==0$$

In[38]:=
```
t[4] = H'[Pi] == 0
```
Out[38]=
$$-4\,i\,h[-4]+3\,i\,h[-3]-2\,i\,h[-2]+i\,h[-1]-i\,h[1]+2\,i\,h[2]-3\,i\,h[3]+4\,i\,h[4]-5\,i\,h[5]+6\,i\,h[6]-7\,i\,h[7]==0$$

In[39]:=
```
t[4] = Map[Expand[I #] &, %]
```
Out[39]=
$$4\,h[-4]-3\,h[-3]+2\,h[-2]-h[-1]+h[1]-2\,h[2]+3\,h[3]-4\,h[4]+5\,h[5]-6\,h[6]+7\,h[7]==0$$

In[40]:=
```
t[5] = H''[0] == 0
```
Out[40]=
$$-16\,h[-4]-9\,h[-3]-4\,h[-2]-h[-1]-h[1]-4\,h[2]-9\,h[3]-16\,h[4]-25\,h[5]-36\,h[6]-49\,h[7]==0$$

In[41]:=
```
t[6] = H''[Pi] == 0
```
Out[41]=
$$-16\,h[-4]+9\,h[-3]-4\,h[-2]+h[-1]+h[1]-4\,h[2]+9\,h[3]-16\,h[4]+25\,h[5]-36\,h[6]+49\,h[7]==0$$

In[42]:=
```
t[7] = H'''[0] == 0
```
Out[42]=
$$64\,i\,h[-4]+27\,i\,h[-3]+8\,i\,h[-2]+i\,h[-1]-i\,h[1]-8\,i\,h[2]-27\,i\,h[3]-64\,i\,h[4]-125\,i\,h[5]-216\,i\,h[6]-343\,i\,h[7]==0$$

In[43]:=
```
t[7] = Map[Expand[I #] &, %]
```
Out[43]=
$$-64\,h[-4]-27\,h[-3]-8\,h[-2]-h[-1]+h[1]+8\,h[2]+27\,h[3]+64\,h[4]+125\,h[5]+216\,h[6]+343\,h[7]==0$$

In[44]:=
```
t[8] = H'''[Pi] == 0
```
Out[44]=
$$64\,i\,h[-4]-27\,i\,h[-3]+8\,i\,h[-2]-i\,h[-1]+i\,h[1]-8\,i\,h[2]+27\,i\,h[3]-64\,i\,h[4]+125\,i\,h[5]-216\,i\,h[6]+343\,i\,h[7]==0$$

In[45]:=
```
t[8] = Map[Expand[I #] &, %]
```
Out[45]=
$$-64\,h[-4]+27\,h[-3]-8\,h[-2]+h[-1]-h[1]+8\,h[2]-27\,h[3]+64\,h[4]-125\,h[5]+216\,h[6]-343\,h[7]==0$$

Elimination using the low-pass conditions

In[46]:=
```
S1 = Solve[Table[t[k], {k, 1, 8}], Table[h[k], {k, -4, 3}]]
```
Out[46]=
$$\Big\{\Big\{h[-4]\to h[4]+4\,h[6],\ h[-3]\to \frac{1}{32}\left(-\sqrt{2}+32\,h[5]+128\,h[7]\right),$$
$$h[-2]\to -4\,h[4]-15\,h[6],\ h[-1]\to \frac{1}{32}\left(9\sqrt{2}-128\,h[5]-480\,h[7]\right),$$
$$h[0]\to \frac{1}{2}\left(\sqrt{2}+12\,h[4]+40\,h[6]\right),\ h[1]\to \frac{1}{32}\left(9\sqrt{2}+192\,h[5]+640\,h[7]\right),$$
$$h[2]\to -2\,(2\,h[4]+5\,h[6]),\ h[3]\to \frac{1}{32}\left(-\sqrt{2}-128\,h[5]-320\,h[7]\right)\Big\}\Big\}$$

Solving the non-linear equations (orthogonality)

In[47]:=
```
S2 = Table[o[k], {k, 1, 6}] /. S1[[1]];
```

In[48]:=

```
S2 = Simplify[%]
```

Out[48]=

$$\left\{\frac{105}{128} + 6\sqrt{2}\,h[4] + 70\,h[4]^2 + \frac{21\,h[5]}{8\sqrt{2}} + 70\,h[5]^2 + 20\sqrt{2}\,h[6] + 448\,h[4]\,h[6] + 742\,h[6]^2 + \frac{51\,h[7]}{8\sqrt{2}} + 448\,h[5]\,h[7] + 742\,h[7]^2 == 1,\right.$$

$$4\sqrt{2}\,h[4] + 56\,h[4]^2 + \frac{3\,h[5]}{4\sqrt{2}} + 56\,h[5]^2 + \frac{25\,h[6]}{\sqrt{2}} + 350\,h[4]\,h[6] + 560\,h[6]^2 + \frac{7\,h[7]}{8\sqrt{2}} + 350\,h[5]\,h[7] + 560\,h[7]^2 == \frac{63}{512},$$

$$28\,h[4]^2 + 28\,h[5]^2 + 2\sqrt{2}\,h[6] + 220\,h[6]^2 + h[4]\left(\sqrt{2} + 160\,h[6]\right) + 220\,h[7]^2 == \frac{9}{256} + \frac{5}{8}\,h[5]\left(\sqrt{2} - 256\,h[7]\right) + \frac{15\,h[7]}{4\sqrt{2}},$$

$$\frac{1}{512} + \frac{3\,h[5]}{4\sqrt{2}} + \frac{h[6]}{\sqrt{2}} + \frac{15\,h[7]}{16\sqrt{2}} - 35\,h[5]\,h[7] == 8\,h[4]^2 + 8\,h[5]^2 + 35\,h[4]\,h[6] + 20\,h[6]^2 + 20\,h[7]^2,$$

$$h[4]^2 + h[5]^2 + \frac{9\,h[7]}{16\sqrt{2}} == \frac{h[5]}{16\sqrt{2}} + 15\left(h[6]^2 + h[7]^2\right),$$

$$\left.h[4]\,h[6] + 4\,h[6]^2 + h[7]\,(h[5] + 4\,h[7]) == \frac{h[7]}{16\sqrt{2}}\right\}$$

In[49]:=

```
sol = NSolve[S2, Table[h[k], {k, 4, 7}]]
```

Out[49]=

```
{{h[4] → -0.327762, h[5] → 0.136027, h[6] → 0.0765196, h[7] → -0.034858},
 {h[4] → -0.0204979, h[5] → 0.0788352, h[6] → -0.00207822, h[7] → -0.00627469},
 {h[4] → 0.0500235, h[5] → 0.0248043, h[6] → -0.0128456, h[7] → 0.00119457},
 {h[4] → 0.0236802, h[5] → 0.00561143,
  h[6] → -0.00182321, h[7] → -0.000720549}}
```

In[50]:=

```
c12coeffs = Table[h[k], {k, -4, 7}] /. S1[[1]] /. sol[[4]]
```

Out[50]=

```
{0.0163873, -0.0414649, -0.0673726, 0.38611, 0.812724, 0.417005, -0.0764886,
 -0.0594344, 0.0236802, 0.00561143, -0.00182321, -0.000720549}
```

Frequency representations

In[51]:=

```
C12[ω_] := Sum[c12coeffs[[k]] Exp[I ω (k - 5)], {k, 1, 12}]
```

In[52]:=

```
C12[ω]
```

Out[52]=

$$0.812724 + 0.38611\,e^{-i\omega} + 0.417005\,e^{i\omega} - 0.0673726\,e^{-2i\omega} - 0.0764886\,e^{2i\omega} - 0.0414649\,e^{-3i\omega} - 0.0594344\,e^{3i\omega} + 0.0163873\,e^{-4i\omega} + 0.0236802\,e^{4i\omega} + 0.00561143\,e^{5i\omega} - 0.00182321\,e^{6i\omega} - 0.000720549\,e^{7i\omega}$$

In[53]:=

```
Plot[{Abs[C6[ω]], Abs[C12[ω]]}, {ω, 0, Pi}, Filling → Axis]
```

Out[53]=